
PROBLEM 1.4-3: Fuel sphere (revisited) 
Reconsider Problem 1.3-7 using a numerical model developed in EES.   
a.) Plot the temperature as a function of position within the fuel. 
 
The inputs are entered in EES and a function is defined to return the volumetric generation. 
 
$UnitSystem SI MASS RAD PA  K J 
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in 
 
function gv(gve,r,r_fuel,b) 
" Inputs: 
gve - volumetric generation at the edge (W/m^3) 
r - radius (m) 
r_fuel - radius of fuel element (m) 
b - exponent (-) 
 
Outputs: 
gv - volumetric rate of thermal energy generation (W/m^3)" 
 
 gv=gve*(r/r_fuel)^b 
 
end 
 
"Inputs" 
r_fuel=5[cm]*convert(cm,m) "radius of fuel" 
k_fuel=1 [W/m-K] "conductivity of fuel" 
r_clad=7[cm]*convert(cm,m) "cladding radius" 
k_clad=300 [W/m-K] "cladding conductivity" 
h_gas=100 [W/m^2-K] "convection coefficient" 
T_gas=converttemp(C,K,500) "gas temperature" 
gve=5e5 [W/m^3] "generation at the center" 
b=1 [-] "decay constant" 
 
The nodal positions are specified: 
 
N=11 [-]   "number of nodes" 
duplicate i=1,N 
 r[i]=(i-1)*r_fuel/(N-1) "location of each node" 
end 
Dr=r_fuel/(N-1) "distance between nodes" 
 
Energy balances on the internal nodes lead to: 
 
"internal nodes" 
duplicate i=2,(N-1) 
 g_dot[i]=4*pi*r[i]^2*Dr*gv(gve,r[i],r_fuel,b) "generation" 
 q_dot_LHS[i]=4*pi*(r[i]-Dr/2)^2*k_fuel*(T[i-1]-T[i])/Dr "conduction from node i-1" 
 q_dot_RHS[i]=4*pi*(r[i]+Dr/2)^2*k_fuel*(T[i+1]-T[i])/Dr "conduction from node i+1" 
 g_dot[i]+q_dot_LHS[i]+q_dot_RHS[i]=0 "energy balance" 
end 
 
An energy balance on node N leads to: 



 
"surface node" 
Rst_clad=(1/r_fuel-1/r_clad)/(4*pi*k_clad) "conduction resistance of cladding" 
Rst_conv=1/(4*pi*r_clad^2*h_gas) "convection resistance" 
g_dot[N]=4*pi*r[N]^2*Dr*gv(gve,r[N],r_fuel,b)/2 "generation" 
q_dot_LHS[N]=4*pi*(r[N]-Dr/2)^2*k_fuel*(T[N-1]-T[N])/Dr "conduction from node N-1" 
q_dot_RHS[N]=(T_gas-T[N])/(Rst_clad+Rst_conv) "heat transfer from gas" 
g_dot[N]+q_dot_LHS[N]+q_dot_RHS[N]=0 "energy balance" 
 
An energy balance on node 1 leads to: 
 
"inner node" 
g_dot[1]=4*pi*(Dr/2)^3*gv(gve,r[1],r_fuel,b)/2/3 "generation" 
q_dot_RHS[1]=4*pi*(r[1]+Dr/2)^2*k_fuel*(T[2]-T[1])/Dr "conduction from node 2" 
g_dot[1]+q_dot_RHS[1]=0 "energy balance" 
 
The solution is converted to °C. 
 
duplicate i=1,N 
 T_C[i]=converttemp(K,C,T[i]) 
end 
 
The temperature as a function of position is shown in Figure 1. 

Figure 1: Temperature as a function of radial position, predicted by the numerical model and the analytical 
model derived in Problem 1.3-7. 
 
b.) Verify that your answer agrees with the analytical solution obtained in Problem 1.3-7. 
 
The analytical solution from Problem 1.3-7 is evaluated at the same radial locations used in the 
numerical model: 
 
"Boundary condition expressions" 
dTdr_rfuel =-2*r_fuel*gve/k_fuel/(b^2+6+5*b)-r_fuel*gve/k_fuel/(b^2+6+5*b)*b  "from Maple" 



T_rfuel =-r_fuel^2*gve/k_fuel/(b^2+6+5*b)+C2 "from Maple" 
-4*pi*r_fuel^2*k_fuel*dTdr_rfuel=(T_rfuel-T_gas)/(Rst_clad+Rst_conv)"boundary condition" 
 
"Solution" 
duplicate i=1,N 
 T_an[i]=-r[i]^2*gve/k_fuel/(b^2+6+5*b)*(r[i]/r_fuel)^b+C2 "solution from Maple" 
 T_an_C[i]=converttemp(K,C,T_an[i]) 
end 
 
The analytical solution is overlaid onto the numerical result in Figure 1. 
 
c.) Plot some aspect of the solution as a function of the number of nodes used in the numerical 

model and determine the number of nodes required for an accurate solution. 
 
Figure 2 illustrates the maximum temperature in the fuel element as a function of the number of 
nodes and shows that at least 50 nodes is required to obtain an accurate solution. 

Figure 2: Maximum temperature in the fuel element as a function of the number of nodes. 

 
 
 


